
Généralités sur la programmation VBA sous Excel

R.R. – Université Lyon 2

Programmation sous Excel via VBA (Visual Basic pour Applications)

Fonctions personnalisées.
Complètement standardisée. Valable pour les 
autres classeurs et même, si pas d’accès aux 
objets spécifiques d’Excel, pour les autres 
outils Office.

Macros.
Manipulation directe des objets Excel (classeurs, 
feuilles, cellules, graphiques, etc.)

Programmation de tâches
Programmation de séquences d’actions plus 
ou moins complexes, faisant intervenir ou 
non des mécanismes algorithmiques.

Interfaces graphiques.
Boîtes de dialogues standards. Mais aussi les formulaires 
personnalisées pour faciliter les accès aux fonctionnalités. 
Nécessite une certaine formalisation et la connaissance 
des principes de la programmation évènementielle.

Enregistreur de macros
Transformation de séquences d’action en 
programme VBA. Ne nécessite pas la 
connaissance de la programmation, mais peu de 
souplesse (structure fixe, peu adaptable…)

Programmation des macros
Très puissant. Nécessite la connaissance des 
principes de la programmation et de la syntaxe 
d’accès aux objets Excel.

Points importants. Connaissance de l’algorithmie, langage de programmation Visual 

Basic. Les instructions sont écrites dans des fonctions (function) et procédures (sub), 

qui sont regroupées dans des modules. Nous travaillons dans VBE (Visual Basic Editor).

https://fr.wikipedia.org/wiki/Visual_Basic_for_Applications

ALGORITHMIE - PROGRAMMATION

Généralités sur la programmation

R.R. – Université Lyon 2

Algorithmie vs. Programmation

Algorithmie

• Solution « informatique » relative à un 

problème

• Suite d’actions (instructions) 

appliquées sur des données

• 3 étapes principales : 

1. saisie (réception) des données

2. Traitements

3. restitution (application) des résultats

Programme

• Transcription d’un algorithme avec une 

syntaxe prédéfinie

• Visual Basic pour Applications

• Même principes fondamentaux que les 

autres langages objets (Java, C#, etc.)

• VBA agit en interaction avec les fonctions 

prédéfinies disponibles dans la suite Office

Mode compilé vs. Mode interprété

Langage interprété : + portabilité application ; - lenteur (R, VBA, Python…)

Langage compilé : + rapidité ; - pas portable

(solution possible : write once, compile anywhere ; ex. Lazarus)

Langage pseudo-compilé : + portabilité plate-forme ; - lenteur (?)

(principe : write once, run anywhere ; ex. Java et le principe JIT)

VBA (Visual Basic pour Applications) est un langage de programmation 

dédié principalement aux applications Microsoft Office. Il est basé sur le 

langage Visual Basic, mais ne peut s’exécuter que dans une application 

hôte Microsoft Office, et non de manière autonome.

https://fr.wikipedia.org/wiki/Visual_Basic_for_Applications
https://fr.wikipedia.org/wiki/Visual_Basic

Etapes de la conception d’un programme (une application)

1. Déterminer les besoins et fixer les objectifs : que doit faire le logiciel, dans quel cadre va-

t-il servir, quels seront les utilisateurs types ? On rédige un cahier des charges avec le 

commanditaire du logiciel (Remarque : commanditaire = maître d’ouvrage ; réalisateur = 

maître d’œuvre)

2. Conception et spécifications : quels sont les fonctionnalités du logiciel, avec quelle 

interface ?

3. Programmation : modélisation et codage

4. Tests : obtient-on les résultats attendus, les calculs sont corrects, y a-t-il plantage et dans 

quelles circonstances ? (tests unitaires, tests d’intégration, etc.)

5. Déploiement : installer le chez le client (vérification des configurations, installation de 

l’exécutable et des fichiers annexes, etc.)

6. Maintenance : corrective, traquer les bugs et les corriger (patches) ; évolutive (ajouter 

des fonctionnalités nouvelles au logiciel : soit sur l’ergonomie, soit en ajoutant de 

nouvelles procédures)

LANGAGE VISUAL BASIC

Programme : suite d’instructions manipulant des données

R.R. – Université Lyon 2

Visual Basic possède tous les attributs d’un langage de programmation

Données typées. Visual Basic propose les types usuels de la programmation : 
entier, réels, booléens, chaîne de caractères.

Structures avancées de données. Gestion des collections de valeurs 
(énumérations, tableaux) et des objets structurés (enregistrements, classes)

Séquences d’instructions, c’est la base même de la programmation, pouvoir 
écrire et exécuter une série de commandes sans avoir à intervenir entre les 
instructions.

Structures algorithmiques : les branchements conditionnels et les boucles.

Les outils de la programmation structurée : pouvoir regrouper du code dans des 
procédures et des fonctions. Organisation du code en modules et possibilité de 
distribuer ces dernières.

Visual Basic n’est pas « case sensitive », il ne différencie pas les termes écrits en 
minuscule et majuscule.

Type de données

Le type de données définit le type d’opérateurs qu’on peut leur appliquer.

• Numérique qui peut être réel (double) ou entier (long). Les opérateurs 
applicables sont : +, -, *, / (division réelle), \ (division entière), mod (modulo)

Exemple : 5 / 2  2.5 ; 5 \ 2  2 ; 5 mod 2  1

• Booléen (boolean) qui ne prend que deux valeurs possibles : True et False. Les 
opérateurs sont : not, and, or.

Exemple : True and False  False

• Chaîne de caractères (suite) qui correspond à une suite de caractères délimitée par des 
guillemets ‘’ ’’. Les opérateurs possibles sont la concaténation, la suppression d’une sous-
partie, la copie d’une sous-partie, etc.

Exemple : ‘’toto’’ est une chaîne de caractères, toto on ne sait pas ce que c’est (pour l’instant)

Habituellement, les opérations font intervenir des données de type identique 

et renvoie un résultat du même type.

Opérateurs de comparaison

Les opérateurs de comparaison confrontent des données de même type, 
mais le résultat est un booléen

= ; <> ; > ; >= ; < ; <=
égalité

différent Supérieur strict, 
supérieur ou égal

inférieur strict, 
inférieur ou égal

Exemples

5 > 2  True

5 > ‘’toto’’  illicite

5 <> 5  False

‘’toto’’ > ‘’tata’’  True

Licite. Comparaison de gauche à droite 
basée sur le code ASCII. Arrêt des 
comparaisons dès qu’indécision levée.

Variables et premières instructions

R.R. – Université Lyon 2

Les variables correspondent à des identifiants auxquels sont associés des valeurs d’un type 
donné. Elles matérialisent un espace mémoire avec un contenu que l’on peut lire ou écrire.

Déclaration 
d’une variable

Dim v As Double

Dimensionner. Réserver 
un espace mémoire .

Indication 
du type

Type associé à 
la variable

Identifiant de la variable, utilisable 
dans la suite du programme

Affectation. Attribuer 
une valeur à la variable

Opération et 
affectation

v = 2.5

= est le symbole d’affectation. A gauche de =
on modifie le contenu dans une variable, à 
droite on lit le contenu d’une variable. C’est 
pour cette raison que l’instruction v = v + 1
est licite.

x = v * 2 La valeur 5 est écrite dans la variable 
x qui doit être déclarée au préalable.

https://fr.wikipedia.org/wiki/Variable_(informatique)

FONCTIONS PERSONNALISÉES

Ecriture et utilisation des fonctions personnalisées dans Excel

R.R. – Université Lyon 2

Programmation des fonctions personnalisées

Une fonction personnalisée est une fonction VBA qui peut appelée 
dans un classeur Excel. Elle prend en entrée des informations en 
provenance des feuilles du classeur (principalement) et renvoie une 
valeur insérée dans une cellule (le plus souvent également).

Formalisme Function NomFonction(paramètres) as type de donnée

Est un identifiant qui doit 
respecter la syntaxe VBA

Les informations que prend en entrée la fonction, elles prennent la 
forme nom_parametre as type de paramètre. Il peut y en avoir 

plusieurs, ils sont séparés par des « , » dans ce cas.

Type de la valeur 
retournée par la fonction.

Un classeur Excel contenant du code VBA doit être enregistré au format 

XLSM, prenant en charge les macros. Sinon on perd son code.

Programmation dans Visual Basic Editor

Entrée : prix HT (réel)
Sortie : prix TTC (réel)

Il faut créer un module pour 
programmer une fonction 
personnalisée

Public pour que la fonction soit visible en dehors 
du module, notamment dans la feuille de calcul

Les lignes de commentaires 
commencent avec une ‘ et sont 
mises automatiquement en vert. 

Utilisation de la fonction dans une feuille Excel

La fonction est insérable dans la feuille de 
calcul comme n’importe quelle autre 
fonction Excel. Elle est accessible dans la 
catégorie « Fonctions personnalisées ».

Le résultat s’affiche une fois la fonction 
insérée et validée. La fonction est 
automatiquement appelée à chaque fois que 
la feuille a besoin d’être recalculée (comme 
pour les autres fonctions standards d’Excel).

Fonction avec plusieurs paramètres

Entrées : prix HT (réel), tva (réel)
Sortie : prix TTC (réel)

'fonction avec 2 paramètres
Public Function MonPrixTTCTva(pht As Double, tva As Double) As Double
'déclaration d'une variable intermédiaire
Dim pttc As Double
'calcul exploitant les paramètres en entrée
pttc = pht * (1 + tva)
'retourner le résultat
MonPrixTTCTva = pttc
End Function

(1) Le séparateur de paramètres est la 
« , » lors de la définition de la fonction.

(2) Très étrangement, il devient « ; » lorsqu’on 
appelle la fonction dans la feuille de calcul.

STRUCTURES ALGORITHMIQUES

Plus loin avec la programmation…

R.R. – Université Lyon 2

Branchement conditionnel IF

Permet d’activer une partie du code en fonction de la 
réalisation d’une condition ou pas.

If condition Then
bloc d’instructions
si la condition est vraie

Else
bloc d’instructions
si la condition est fausse

End If

Sy
n

ta
xe

(1) Condition est souvent une opération de comparaison

(2) La valeur de retour de Condition est de type booléen (True ou False)

(3) Then doit être sur la même ligne que If

(4) La partie Else est facultative (ne rien faire si la condition est fausse)

(5) Il est possible d’imbriquer une autre structure conditionnelle If dans les blocs d’instructions

Branchement conditionnel IF – Un exemple

Entrées : prix HT (réel), catégorie de produit (chaîne)
Sortie : prix TTC (réel)

Public Function MonTTCBis(pht As Double, cat As String) As Double
'déclarer la variable de calcul
Dim pttc As Double
'en fonction de la catégorie de produit
If (cat = "luxe") Then

pttc = pht * 1.33
Else

‘la valeur de cat est différente de ‘’luxe’’
pttc = pht * 1.2

End If
'renvoyer le résultat
MonTTCBis = pttc
End Function

Branchement multiple SELECT CASE

Permet d’activer une partie du code en fonction des valeurs prises par une 
variable de contrôle. Peut se substituer au IF, mais pas toujours, tout dépend 
de la forme de la condition (condition composée, on doit passer par un IF).

Select Case variable
Case valeur 1

bloc d’instructions
Case valeur 2

bloc d’instructions
...
Case Else

bloc d’instructions
End Select

Sy
n

ta
xe

(1) Variable est la variable de contrôle, elle peut être de n’importe quel type en VBA, y compris un 

réel ou une chaîne de caractères

(2) Valeur doit être de type compatible avec variable

(3) La partie Case Else est facultative

(4) L’imbrication avec un autre IF ou un autre Select Case (autre variable de contrôle) est possible.

Branchement multiple SELECT CASE – Un exemple

Entrées : prix HT (réel), catégorie de produit (chaîne)
Sortie : prix TTC (réel)

'fonction select case
Public Function MonTTCSelon(pht As Double, cat As String) As Double
'déclarer la variable de calcul
Dim pttc As Double
'en fonction de la catégorie de produit
Select Case cat

Case "luxe"
pttc = pht * 1.33

Case Else
pttc = pht * 1.2 'toute autre valeur que ‘’luxe’’

End Select
'renvoyer le résultat
MonTTCSelon = pttc
End Function

Branchement multiple SELECT CASE – Plages de valeurs

Il est possible d’introduire des plages de valeurs dans la partie Case de 

la structure Select Case. La comparaison devient plus sophistiquée. 

Variable est un numérique dans ce cas, entier ou même réel.

Select Case variable
Case Is op.de.comparaison valeur

bloc d’instructions

Case valeur de départ To valeur de fin
bloc d’instructions

Case Else
bloc d’instructions

End Select

Sy
n

ta
xe

Branchement multiple SELECT CASE – Plages de valeurs – Un exemple

Entrée : quantité (entier)
Sortie : prix unitaire (réel)
Calcul : quantité < 100  p.u. = 0.5

100  quantité  200  p.u. = 0.3
quantité > 200  p.u. = 0.2

'calcul du prix unitaire en fonction de la quantité
Public Function MonPU(quantite As Long) As Double
'variable intermédiaire
Dim pu As Double
'selon les valeurs de quantité
Select Case quantite

Case Is < 100
pu = 0.5

Case 100 To 200
pu = 0.3

Case Is > 200 'Case Else aurait fait l'affaire aussi
pu = 0.2

End Select
MonPU = pu
End Function

Boucle POUR (FOR)

Faire répéter l’exécution d’un bloc d’instructions. Le nombre d’itérations 

est contrôlé par un indice.

For indice = val.départ to val.fin step pas
bloc d’instructions
...

Next indiceSy
n

ta
xe

(1) Indice est un type ordonné, très souvent un numérique

(2) pas contrôle le passage d’une valeur à l’autre d’indice, si omis, pas = 1 par défaut

(3) Next entérine le passage à la valeur suivante de indice, si cette prochaine valeur 

est > à val.fin, on sort de la boucle

(4) Val.fin doit être superieure à val.départ pour que l’on rentre dans la boucle

(5) Si pas est négatif, val.fin doit être inférieure à val.départ cette fois-ci

(6) L’instruction Exit For permet de sortir prématurément de la boucle

(7) On peut imbriquer des boucles (une boucle à l’intérieur d’une autre boucle)

Boucle FOR – Un exemple

Entrée : n (entier)
Sortie : S (réel)
Calcul : S = 1² + 2² + … + n²

'calcul de la somme des carrés des valeurs
Public Function MaSommeCarre(n As Long) As Double
'variables de calcul (s pour la somme, i : indice)
Dim s As Double, i As Long
'initialisation
s = 0
'boucle avec l'indice i
For i = 1 To n Step 1

s = s + i ^ 2
‘Next joue le rôle de l’incrémentation (i suivant)
Next i
'renvoyer le résultat
MaSommeCarre = s
End Function

Boucle TANT QUE… FAIRE (DO WHILE…LOOP)

Faire répéter l’exécution d’un bloc d’instructions. Le nombre d’itérations 

est contrôlé par une condition. Attention à la boucle infinie c.-à-d. la 

condition permettant de sortir de la boucle n’est jamais déclenchée.

Si la condition est fausse d’emblée. On 

peut ne pas rentrer dans la boucle.

Do While condition
Bloc d’instructions...
...

LoopSy
n

ta
xe

(1) Condition est un booléen, c’est souvent une opération de comparaison

(2) On continue l’exécution TANT QUE la condition est vraie ; si la condition 

est fausse, on sort de la boucle

(3) Exit Do permet de provoquer la sortie prématurée de la boucle



27

Boucle DO WHILE…LOOP (un exemple)

Entrée : n (entier)
Sortie : S (réel)
Calcul : S = 1² + 2² + … + n²

'calcul de la somme des carrés des valeurs
Public Function MaSommeCarreWhile(n As Long) As Double
'variables de calcul
Dim s As Double, i As Long
'initialisation
s = 0
'il nous revient aussi d'initialiser l'indice
i = 1
'boucle TANT QUE
Do While (i <= n)

'sommer
s = s + i ^ 2
'pas de next, nous devons incrémenter l'indice
i = i + 1

Loop
'renvoyer le résultat
MaSommeCarreWhile = s
End Function

Boucle FAIRE…TANT QUE (DO…LOOP WHILE)

Faire répéter l’exécution d’un bloc d’instructions. Le nombre d’itérations 

est contrôlé par une condition.

On est sûr de rentrer au moins une fois dans la boucle.

Do
Bloc d’instructions
...
...

Loop While condition

Sy
n

ta
xe

Le choix de la bonne structure (Faire.. Tant Que ou Tant Que.. 

Faire) dépend du problème à traiter

Les variantes des boucles DO

Les boucles DO contrôlées par une condition sont très riches en VBA.

Do { While | Until } condition

[ statements ] 

[ Exit Do ] 

[ statements ]

Loop

-or-

Do

[ statements ] 

[ Exit Do ] 

[ statements ]

Loop { While | Until } condition

Le Répeter… Jusqu’à (Until) existe aussi.

https://msdn.microsoft.com/fr-fr/library/eked04a7(v=vs.90).aspx

LE TYPE RANGE

Le type « plage de cellules » spécifique à Excel

R.R. – Université Lyon 2

Le type RANGE

Le type RANGE désigne une plage de cellules, c’est un type spécifique à Excel.

Ex
em

p
le

Un bloc de cellules (B3:D4) est passé en 
paramètre de la fonction. Ce bloc est forcément 
rectangulaire, avec, ici : 2 lignes et 3 colonnes.

Coin en haut et à gauche de la plage de cellules passée en paramètre de la fonction = 
coordonnée (1, 1) c.-à-d. ligne n°1 et colonne n°1, quelle que soit la position absolue 
de la plage dans la feuille de calcul (ici le coin nord-ouest est en B3)

La fonction MaSommeRange() est censée faire la même chose 

que la fonction standard SOMME() d’Excel.

Exploiter le type Range en VBA

Entrée : plage (range)
Sortie : S (réel)
Calcul : Somme des valeurs

'Travail sur le type Range

Public Function MaSommeRange(plage As Range) As Double

'variables intermédiaires

Dim s As Double, i As Long, j As Long

'initialisation de la somme

s = 0

'parcours de la plage de cellules

For i = 1 To plage.Rows.Count Step 1 'lignes

For j = 1 To plage.Columns.Count Step 1 'colonnes

'lecture des valeurs et somme

s = s + plage.Cells(i, j).Value

Next j

Next i

'renvoyer le résultat

MaSommeRange = s

End Function

Nombre de lignes de 
la plage de cellules.

Nombre de 
colonnes.

Accès à la valeur (Value) de la 
cellule : ligne n°i, colonne n°j

Lignes et colonnes commencent à l’indice 1, quelle 
que soit la position de la plage dans la feuille.

Une boucle adaptée pour les plages de cellules – For Each

R.R. – Université Lyon 2

La boucle For Each est adaptée au parcours des collections. 

Or une plage de cellules est une collection de cellules.

'Travail sur le type Range avec un For Each

Public Function MaSommeRangeEach(plage As Range) As Double

'variables intermédiaires

Dim s As Double, cellule As Range

'initialisation de la somme

s = 0

'parcours de la plage de cellules

For Each cellule In plage

s = s + cellule.Value

Next cellule

'renvoyer le résultat

MaSommeRangeEach = s

End Function

Une cellule est une plage de 
cellules avec une seule cellule.

plage fait figure de collection à traiter. 
Qu’importe le sens du parcours ici (ligne 
par ligne, ou colonne par colonne).

C’est bien la valeur contenue dans la 
cellule qui est exploitée pour la somme.

https://msdn.microsoft.com/fr-fr/library/5ebk1751(v=vs.90).aspx

LE TYPE VARIANT

Type spécial qui peut contenir toutes sortes de valeur

R.R. – Université Lyon 2

Le type Variant

Le type de variant peut gérer tout type de valeurs. Il est très souple, 

particulièrement commode quand on ne connaît pas à l’avance le type à 

utiliser. Mais attention, il ne faut pas en abuser, il est très lent parce que 

multiplie les vérifications à chaque accès à la variable correspondante.

Entrée : a, b (réel)
Sortie : a/b si b  0, « division par zéro » sinon

Un coup, la fonction renvoie un réel, un 
autre coup elle doit renvoyer une chaîne 
de caractères.

'utilisation du type variant

Public Function MaDivision(a As Double, b As Double) As Variant

'var. intermédiaire

Dim resultat As Variant

'calcul

If (b <> 0) Then

resultat = a / b

Else

resultat = "division par zéro"

End If

'renvoyer le résultat

MaDivision = resultat

End Function

Dans la même variable resultat, de type 
variant, on peut affecter un réel et une 
chaîne de caractères.

Le type Variant est vraiment très souple

On peut s’en servir pour renvoyer un tableau. Une fonction peut donc renvoyer 

plusieurs valeurs d’un coup, à l’instar des fonctions matricielles d’Excel (il faut valider 

la saisie de la fonction avec la séquence de touches CTRL + MAJ + ENTREE).

'renvoyer plusieurs valeurs

Public Function MonMinMax(a As Double, b As Double) As Variant

'un tableau interne - matrice 2 lignes et 1 colonne

Dim tableau(1 To 2, 1 To 1) As Double

'identifier le min et le max

If (a < b) Then

tableau(1, 1) = a

tableau(2, 1) = b

Else

tableau(1, 1) = b

tableau(2, 1) = a

End If

'renvoyer le tableau

MonMinMax = tableau

End Function

On a bien une fonction matricielle 
comme peuvent en témoigner les 
accolades { } qui encadrent l’appel 
de la fonction.

LES MACROS (1)

Programmation des macros – Travailler directement sur les feuilles

R.R. – Université Lyon 2

Macros ?

Les macros sont également des procédures que l’on crée à l’intérieur d’un module. 

Mais, à la différence des Function, ce sont des Sub() sans paramètres qui peuvent 

manipuler (accéder et modifier) directement les objets Excel (classeurs, feuilles, 

cellules, graphiques, scénarios, tableaux croisés dynamiques…).

Ils ne s’exécutent pas de la même manière. Au lieu de les insérer dans une cellule, ils 

se lancent globalement via le bouton MACROS dans le ruban DEVELOPPEUR.

Enregistreur de macros

Une manière simple de générer une macro est de lancer l’enregistreur de macros. 

Du code VBA est automatiquement généré.

Exemple : mettre en gras et vert 
le contenu des cellules A1 et A2

Un nouveau module 
« Module1 » est 
automatiquement créé.

Une macro correspond à 
un Sub() sans paramètres.

Ecrire du code VBA pour les macros 
se révélera plus simple au final.

https://support.office.com/fr-fr/article/Enregistrer-une-macro-24a026ef-3145-4bf8-a5f2-2fc7889ff74a


40

Enregistreur de macros - Bilan

Avantages :

• Il n’y a pas plus simple pour produire du code, on peut créer et exécuter une macro 

sans aucune notion de programmation

• Il nous donne des indications précieuses sur les commandes associées aux objets 

Excel

Inconvénients :

• On travaille à structure fixée, si la configuration de la feuille change, il n’est pas 

possible de lancer la macro

• On ne bénéficie de la puissance des structures algorithmiques

En définitive :

• Il peut nous aider à rédiger notre code en nous donnant des pistes sur la syntaxe 

des commandes et les objets adéquats à manipuler (ex. imprimer automatiquement 

des feuilles, on lance l’enregistreur une fois, on intègre son code dans le notre à 

l’intérieur d’une boucle).

Ecriture des macros – Les trois principaux objets

Ecrire directement des macros est simple une fois assimilé la philosophie de 

l’approche, et identifié les principaux objets et l’accès à leurs propriétés et méthodes 

(l’enregistreur peut nous y aider).

Classeurs

Feuilles

Cellules

Workbooks(‘’classeur1.xlsm’’).Activate

Activer (sélectionner) le 
classeur dont le nom de 
fichier est ‘’classeur1.xlsm’’

Sheets(‘’Feuil1’’).Activate

Dans le classeur courant, activer la feuille de calcul 
dont le nom est ‘’Feuil1’’ (visible dans la languette 
au bas de la feuille)

Workbooks(‘’classeur1.xlsm’’).Sheets(‘’Feuil1’’).Activate On peut combiner 
les écritures.

Cells(1,1).Value = 15
Dans la feuille courante du classeur courant, insérer 
la valeur 15 dans la cellule ligne n°1, colonne n°1 
c.-à-d. en A1, les coordonnées sont absolues ici.

Sheets(‘’Feuil1’’).Cells(1,1).Value = 15 De nouveau, on peut combiner.

Exemple de macros – Simulation valeurs de TVA

R.R. – Université Lyon 2

Ecrire une macro qui insère différentes valeurs de 

TVA en B2 et récupère les valeurs de prix TTC en B3. Les différentes valeurs de TVA testées doivent être 
retranscrites au fur et à mesure dans la colonne D.

Les valeurs de Prix TTC correspondantes 
doivent être recensées en colonne E 

Sub SimulationTVA()
'variables
Dim pht As Double, pttc As Double
Dim tva As Double
Dim i As Long
'début d'écriture des valeurs en ligne 2
i = 2
'récupérer la valeur du PHT
pht = Cells(1, 2).Value 'en B1
'faire varier la tva de 10% à 30% avec un pas de 5%
For tva = 10 To 30 Step 5

'insérer la valeur de la TVA en B2
Cells(2, 2).Value = tva
'Récupérer le prix ttc en B3
pttc = Cells(3, 2).Value
'inscription des valeurs
'TVA en colonne D
Cells(i, 4).Value = tva
'PTTC en colonne E
Cells(i, 5).Value = pttc
'passage à la ligne suivante
i = i + 1

Next tva
End Sub

A l’issue de la simulation…

Remarque : il faut être sur la feuille 
adéquate avant de lancer la macro, 
sinon le programme ne saura pas où 
chercher Cells(…).

LES MACROS (2)

Travailler sur les sélections de l’utilisateur

R.R. – Université Lyon 2

Sélection simple

Comment programmer une macro qui manipule directement une plage de cellules 

sélectionnée par l’utilisateur ? Attention, nous ne sommes pas dans la même 

configuration que les fonctions personnalisées ici, nous n’insérons pas un résultat 

dans une cellule, nous manipulons et modifions directement la plage sélectionnée.

Exemple : dans cette sélection (les cellules 

doivent être sélectionnées avant de lancer 

la macro !), mettre en police verte les 

cellules contenant une valeur paire.

Sub MesValeursPaires()
'variable intermédiaire
Dim cellule As Range
'boucler sur la sélection
For Each cellule In Selection

'tester le contenu
If (cellule.Value Mod 2 = 0) Then

'modifier la couleur de la police
cellule.Font.ColorIndex = 4

End If
Next cellule
End Sub

Selection est un objet Excel (Selection
est donc un mot clé). Il est de type 
Range (que nous connaissons bien).

10 15 20 13

36 7 8 28

Résultat…

Sélection simple – On aurait pu écrire…

Sub MesValeursPairesBis()

'variables intermédiaires

Dim i As Long, j As Long

'boucler sur les lignes

For i = 1 To Selection.Rows.Count

'boucler sur les colonnes

For j = 1 To Selection.Columns.Count

'tester le contenu

If (Selection.Cells(i, j).Value Mod 2 = 0) Then

'modifier la couleur de la police

Selection.Cells(i, j).Font.ColorIndex = 4

End If

Next j

Next i

End Sub

Aucun doute, Selection est 
bien de type Range

Dans une sélection, de nouveau les 
coordonnées sont relatives c.-à-d. le coin en 
haut à gauche d’une sélection correspond à la 
cellule (ligne n°1, colonne n°1) quelle que soit la 
position de la sélection dans la feuille.

Sélection simple – Un second exemple

Identifier la première cellule contenant la valeur 

minimale dans une plage, mettre sa police en bleu.

10 15 20 13

36 7 8 28

Sub MonMinBleu()
‘variables intermédiaires
‘min va servir de cellule témoin
Dim cellule As Range, min As Range
'initialisation du témoin sur la 1ère cellule
Set min = Selection.Cells(1, 1)
'parcourir
For Each cellule In Selection

'comparer avec le contenu de la cellule témoin

If (cellule.Value < min.Value) Then
'màj de la cellule témoin
Set min = cellule

End If
Next cellule
'mettre la couleur pour la cellule minimale
min.Font.ColorIndex = 5
End Sub

Range est un objet. Une 
affectation pour une variable 
objet doit être réalisée à 
l’aide de l’instruction Set

Sélections multiples

Une sélection peut être multiple aussi c.-à-d. contenant plusieurs ‘’zones’’

Un exemple de sélection 

multiple avec 3 zones.

Très curieusement, le même mot clé Selection peut être exploité.

Selection.Areas.Count

Selection.Areas(k)

Nombre de ‘’zones’’ dans la sélection.

Accès à la zone n°k (qui est de type Range). 
Areas est une collection de zones.

Sélection multiple – Un exemple

R.R. – Université Lyon 2

Pour chaque zone, mettre en 

police bleue la cellule contenant 

la valeur minimale.

Sub MonMinZoneBleu()

'var. intermédiaires

Dim zone As Range, min As Range

'pour chaque zone

For Each zone In Selection.Areas

'à l'intérieur de chaque zone

'initialisation

Set min = zone.Cells(1, 1)

'parcours des cellules

For Each cellule In zone

'comparer

If (cellule.Value < min.Value) Then

'màj de la variable témoin

Set min = cellule

End If

Next cellule

'mettre la couleur pour la cellule minimale

min.Font.ColorIndex = 5

'passage à la zone suivante

Next zone

End Sub

Selection.Areas est une collection. On peut 
utiliser un For Each. On aurait pu aussi passer 
par un accès indicé. Par ex.
For k = 1 to Selection.Areas.Count

Set zone = Selection.Areas(k)

Etc…

Résultat…

BOÎTES DE DIALOGUE

Boîtes de dialogue standards

Les boîtes de dialogue permettent d’interagir avec l’utilisateur. Nous nous en 

tenons aux plus simples ici. InputBox() pour la saisie, MsgBox() pour l’affichage.

Sub MesBoitesDeDialogue()

'var. intermédiaire

Dim prenom As String

'saisie

prenom = InputBox("Entrer votre prénom", "Saisie", "")

'affichage

MsgBox ("Bonjour " & prenom)

End Sub

InputBox() est une fonction

Noter la concaténation de chaînes de caractères pour 
faire apparaître le prénom dans la boîte de dialogue.

EXPLOITER LES FONCTIONS 
NATIVES D’EXCEL

Accéder aux fonctions natives d’Excel dans nos programmes

Excel dispose de fonctions natives puissantes. Nous 

pouvons y accéder dans nos programmes VBA.

Sub MaMoyenneSelection()

'var. intermédiaire

Dim moyenne As Double

'vérifier la sélection

If (Selection.Areas.Count > 1) Then

MsgBox ("Attention, ce n'est pas une sélection simple")

Else

'faire calculer la moyenne de la sélection par Excel

moyenne = Application.WorksheetFunction.Average(Selection)

MsgBox ("La moyenne est " & Str(moyenne))

End If

End Sub

Exemple : Vérifier qu’une sélection est simple (une 

seule zone), puis calculer et afficher la moyenne 

des valeurs dans une boîte de dialogue.

Noter la syntaxe.

Références

De la documentation à profusion (inutile d’acheter des livres sur VBA)

Site de cours de Microsoft

VBA sous Excel : https://msdn.microsoft.com/fr-fr/library/office/ee814737(v=office.14).aspx

Structures de décision : https://msdn.microsoft.com/fr-fr/library/hh892482(v=vs.90).aspx

Structures de boucle : https://msdn.microsoft.com/fr-fr/library/ezk76t25(v=vs.90).aspx

Autres cours et supports

Le Compagnon Info : http://www.lecompagnon.info/excel/

Excel Easy : http://www.excel-easy.com/

Cours VBA Gratuit : http://www.excel-pratique.com/fr/vba.php

Excel VBA for Complete Beginners : http://www.homeandlearn.org/index.html

Et d’autres très nombreux encore… faites chauffer les moteurs de recherche.

https://msdn.microsoft.com/fr-fr/library/office/ee814737(v=office.14).aspx
https://msdn.microsoft.com/fr-fr/library/hh892482(v=vs.90).aspx
https://msdn.microsoft.com/fr-fr/library/ezk76t25(v=vs.90).aspx
http://www.lecompagnon.info/excel/
http://www.excel-easy.com/
http://www.excel-pratique.com/fr/vba.php
http://www.homeandlearn.org/index.html

